
PROC SUMMARY

(Almost) Everything you need to know about PROC SUMMARY

PROC SUMMARY Overview

▪ Useful for summarizing data overall and/or by categories

▪ Approximately 99% overlap with PROC MEANS

▪ Default output from PROC MEANS is a printed table

▪ Default output from PROC SUMMARY is a SAS Data Set

▪ These defaults can be over-ridden

▪ Can be faster than doing similar calculations in PROC SQL

▪ Computes many useful statistics including:

▪ Mean

▪ Sum

▪ Standard Deviation

▪ N

▪ N Missing

▪ Median

▪ Quartiles and Percentiles

▪ Minimum

▪ Maximum

▪ Range

▪ Median (or other percentiles)

▪ Many other statistics (see SAS Help)

PROC SUMMARY Overview

▪ Data to be used in most examples is SASHELP.CARS (partial view), which you 

all have access to (no LIBNAME statement needed)

PROC SUMMARY Overview

Example 1: Compute average MSRP by Origin

proc summary data=sashelp.cars;

class origin;

var msrp;

output out=summary_out mean=;

run;

Things to note:

▪ The CLASS statement accepts the name of the variable(s) used for categories; can be 
either a numeric or character variable; CLASS is optional but usually present; data does 
not have to be pre-sorted by CLASS variables. PROC SUMMARY figures out how many 
distinct values of each CLASS variable are present (for ORIGIN, there are 3).

▪ The VAR statement specifies the name of the variable(s) to compute statistics from; must 
be a numeric variable; required

▪ The OUTPUT statement requires OUT= followed by the name of the output data set; and 
at least one statistic. If there is no text after MEAN= then the mean gets the same variable 
name as the VAR variable(s). The OUTPUT statement is required. The OUT= is optional 
but recommended; the statistic is also optional but recommended.

PROC SUMMARY Overview

Example 1: Compute average MSRP by Origin

Result of executing this code:

Things to note:

▪ Each value of the CLASS variable gets a row, showing the value of the CLASS 

variable, _TYPE_ (to be explained later), _FREQ_ (number of records found) 

and the mean value of the VAR variable.

▪ You also get a row where the CLASS variable is missing and _TYPE_=0, this is 

the average over all data (regardless of CLASS).

PROC SUMMARY Overview

Example 1 Modified: Compute average MSRP by Origin, but suppose you don’t 

want the first row with the overall average

proc summary data=sashelp.cars nway;

class origin;

var msrp;

output out=summary_out mean=;

run;

Notes: the NWAY option eliminates the overall average row. It also has other effects that do 
not show up in this simple example, these are described later.

PROC SUMMARY Overview

Example 2: Compute average, minimum and maximum of MSRP and Horsepower  
by Origin

proc summary data=sashelp.cars;

class origin;

var msrp horsepower;

output out=summary_out mean=msrp_mean horsepower_mean

min=msrp_min horsepower_min max=msrp_max horsepower_max;

run;

Things to note:

▪ The OUTPUT statement is required; OUT= is used to specify the name of the 
output data set; and at least one statistic is requested. If you have n VAR 
variables, you must specify n variable names after each statistic (use any legal 
SAS variable name).

▪ Except that you can leave the variable names after one statistic blank and the VAR 
variable names would be used in the output data set for that statistic.

▪ Or use output out=summary_out mean= min= max= / autoname;

PROC SUMMARY Overview

Example 2: Compute average, minimum and maximum of MSRP and Horsepower  

by Origin

PROC SUMMARY Overview

Example 3: Compute average using multiple variables in the CLASS Statement

proc summary data=sashelp.cars;

class origin type drivetrain;

var msrp;

output out=summary_out mean=;

run;

PROC SUMMARY Overview

Example 3: Compute average using multiple variables in the CLASS Statement

Overall average (_TYPE_=0)

Average by Drive Train (_TYPE_=1)

Average by Type (_TYPE_=2)

Average by combination of Type and 

Drivetrain (_TYPE_=3)

Average by Origin (_TYPE_=4)

Average by combination of Origin and 

Drivetrain (_TYPE_=5)

Continued on next slide

PROC SUMMARY Overview

Example 3: Compute average using multiple variables in the CLASS Statement

Average of the combination of Origin 

and Type (_TYPE_=6)

Average of the combination of all three 

CLASS variables (_TYPE_=7)

PROC SUMMARY Overview

Example 3 MODIFICATION 1: Compute average using multiple variables in the 

CLASS Statement, but you only want combinations of all three CLASS variables

proc summary data=sashelp.cars nway;

class origin type drivetrain;

var msrp;

output out=summary_out mean=;

run;

Notes: You only get the _TYPE_=7 results 

(i.e. the combination of all three CLASS 

variables). NWAY gives you the 

combination of all CLASS variables

and nothing else.

PROC SUMMARY Overview

Example 3 MODIFICATION 2: You only want results for each CLASS variable by 

itself (and not all those combinations of variables). Use the WAYS command.

proc summary data=sashelp.car;

class origin type drivetrain;

ways 1;

var msrp;

output out=summary_out mean=;

run;

PROC SUMMARY Overview

Example 3 MODIFICATION 2: You only want results for each CLASS variable by 

itself (and not all those combinations of variables); plus the overall average; and 

you want these statistics for MSRP and also for horsepower

proc summary data=sashelp.cars;

class origin type drivetrain;

ways 0 1;

var msrp horsepower;

output out=summary_out mean=;

run;

What would the command ways 2; produce?

PROC SUMMARY Overview

Example 3 MODIFICATION 3: Now compute a new variable which is the delta 

from the overall mean of both MSRP and Horsepower. This can’t be done directly 

in PROC SUMMARY, but it is easily done in a data step.
proc summary data=sashelp.cars;

var msrp horsepower;

output out=summary_out mean=msrp_mean horsepower_mean;

run;

data cars2;

if _n_=1 then set summary_out;

set sashelp.cars;

msrp_delta=msrp-msrp_mean;

horsepower_delta=horsepower-horsepower_mean;

run;

You can also do this in PROC 

STDIZE, but that’s a different 

seminar

PROC SUMMARY Overview

Example 3 MODIFICATION 4: Suppose you want mean by Origin, by DriveTrain

and the combination of Type and DriveTrain. Use the TYPES command instead of 

the WAYS command.

proc summary data=sashelp.cars;

class origin type drivetrain;

types origin drivetrain 

type*drivetrain;

var msrp horsepower;

output out=summary_out mean=;

run;

Hint: in the TYPES command, if you 

want the overall mean, use (for example)

types () origin drivetrain type*drivetrain;

PROC SUMMARY Overview

Example 4: You want the average of OriginalBureau and LTV, both weighted by 

LoanBalance, broken down by vintage. These weighted averages should be 

shown to 2 decimal places.

proc summary data=extract nway;

class vintage;

var originalbureau ltv;

weight loanbalance;

format originalbureau ltv 8.2;

output out=summary_out mean=;

run;

PROC SUMMARY Overview

Example 4 MODIFICATION 1: You want the average of OriginalBureau and LTV, 

both weighted by LoanBalance, broken down by vintage; and you also want the 

sum of the values of LoanBalance by Vintage.

proc summary data=extract nway;

class vintage;

var originalbureau ltv;

weight loanbalance;

format originalbureau ltv 8.2;

output out=summary_out mean= sumwgt=sum_loanbalance;

run;

PROC SUMMARY Overview

Example 4 MODIFICATION 2: You want the average of FICO and LTV, FICO  

weighted by LoanBalance and LTV unweighted, broken down by vintage.

proc summary data=extract nway;

class vintage;

var fico/weight=loanbalance;

var ltv;

format fico ltv 8.2;

output out=summary_out mean=;

run;

PROC SUMMARY Overview

Example 4 Comment:

A weighted average is easy to do in PROC SQL. Partial code:
SUM(a.loanbalance*a.originalbureau)/SUM(a.loanbalance)

For a simple example like this, there probably is very little difference in speed 

doing the analysis in SQL or SUMMARY. If you are computing a lot of statistics in 

a single SQL call, then it is my experience the PROC SUMMARY is faster. 

(Applies to all statistics, not just weighted averages)

If you are slicing the data several different ways, this would require several PROC 

SQL calls, which require SAS to pass through the data several times. But in PROC 

SUMMARY, you need only one PROC SUMMARY call and then you pass through 

the data only once.

PROC SUMMARY Overview

Example 4 Comment:

The results between PROC SQL and PROC SUMMARY do not match in the case 

of a missing value in the numerator of the weighted average. I believe that PROC 

SUMMARY comes up with the proper result, and PROC SQL is wrong. (Applies to 

weighted statistics only, not the MEAN function in SQL)

▪ PROC SUMMARY does not use the record with the missing value in the 

numerator. 

▪ PROC SQL uses the record with the missing value in the denominator, but not 

the numerator. So PROC SQL will use a larger denominator, and essentially 

treats the missing value as a zero in the numerator.

PROC SUMMARY Overview

Example 5: You want an average and median of MSRP by groups of horsepower 

and by origin. Create a format!

proc format;

value horsef low-199='<200' 

200-300='200-300'

301-400='301-400' 

401-high='>400';

run;

proc summary data=sashelp.cars;

class origin horsepower;

var msrp;

output out=summary_out mean= 

median=msrp_median;

format horsepower horsef.;

run;

Why not something like this:
if horsepower<200 then hpower='<200’; else ...

This will not sort properly; formats sort according to the underlying numeric values.

PROC SUMMARY Overview

Example 5 MODIFIED: You want an average of MSRP and Invoice, and the 

median of MSRP (but not the median of invoice) by groups of horsepower and 

origin combined. 

proc format;

value horsef low-200='<200' 

201-300='200-300'

301-400='301-400' 

401-high='>400';

run;

proc summary data=sashelp.cars;

class origin horsepower;

var msrp invoice;

ways 2;

output out=summary_out mean= 

median(msrp)=msrp_median;

format horsepower horsef.;

run;

PROC SUMMARY Overview

Example 6: Determine the maximum value of variable CYCLEDELINQUENCY for 

each loan, keeping identifying information such as PROCESSDATE and 

ORIGINATIONAMOUNT. Use the ID statement to keep the identifying information 

in the output.

proc summary data=extract nway;

class fullaccountnumber;

var cycledelinquency;

output out=summary_out(drop=_:) max=cycledelinquency_max; 

id processdate originationamount;

run;

PROC SUMMARY Overview

Example 6 MODIFIED: Determine maximum value of CYCLEDELINQUENCY for 

each loan, keeping identifying information such as PROCESSDATE and 

ORIGINATIONAMOUNT, and determine the MOB and LOANBALANCE when this 
maximum CYCLEDELINQUENCY occurred. 

proc summary data=shaw_extract nway;

class fullaccountnumber;

var cycledelinquency;

output out=summary_out(drop=_:) max=cycledelinquency_max

maxid(cycledelinquency(mob) cycledelinquency(loanbalance))=

mob_at_max bal_at_max; 

id processdate originationamount;

run;

PROC SUMMARY Overview

Example 7: Computing proportions and weighted proportions. If you have a binary 

(0 or 1) variable, then the mean is the proportion of units that have a 1. 

proc sql;

create table extract as select 

case when cycledelinquency>1 then 1 else 0 end 

as dpd30,processyear,loanbalance

from mydatabase;

quit;

proc summary nway data=extract;

class processyear;

var dpd30;

weight loanbalance;

output out=stuff mean=;

run;

PROC SUMMARY Overview

Example 8: Compute Ever 30 % Delinquent by FICO and PTI bands, table of 

results created by PROC REPORT

First we present the output:

PROC SUMMARY Overview

/* Create FICO and PTI formats that define the buckets */

proc format; 

value ficof 680-699='680-699' 700-719='700-719' 720-739='720-739' 740-779='740-779'

780-9998='780+' 9999='All';

value ptif 0-8='0-8 %' 9-11='9-11 %' 12-14='12-14 %' 15-17='15-17 %' 18-20='18-20 %' 

21-9998='21+ %' 9999='All';

run;

/* Extract data */

proc sql; 

create table deal_review as select 

a.fullaccountnumber,a.processdate,

fico format=ficof.,

pti format=ptif.,

case when a.cycledelinquency>1 then 1 else 0 end as ever30

from mydatabase a;

quit;

PROC SUMMARY Overview

/* Compute statistics, since variable ever30 is binary 0/1, the mean is the percent ever 

30 delinquent */

proc summary data=deal_review;

class fico pti;

var ever30;

output out=_stats_ mean= n=n_loans;

run;

PROC SUMMARY Overview

/* Assign value 9999 to missing fico or missing pti, so it will be formatted as the word 
‘All’ */

data _stats_;

set _stats_;

if missing(fico) then fico=9999;

if missing(pti) then pti=9999;

run;



31

PROC SUMMARY Overview

/* Produce report */

title "Ever 30 Bad Rates by PTI and FICO Range";

title2 "Loans Originated Sep 2015 to August 2016";

proc report data=_stats_;

columns fico pti,(ever30) pti,(n_loans);

define fico/group "FICO" order=internal;

define pti/across "PTI" order=internal;

define ever30/analysis sum "Ever 30 Dq %" format=percent9.2;

define n_loans/analysis sum "N Loans" format=comma8.0;

run;

order = internal forces 

PROC REPORT to order the 

columns and rows in the proper 

numerical order

PROC SUMMARY Overview

What would happen if we used –9999 instead of 9999??
proc format;

value fico 680-699='680-699' 700-719='700-719' 720-739='720-739' 740-779='740-779‘

780-9998='780+' -9999='All';

value ptif 0-8='0-8 %' 9-11='9-11 %' 12-14='12-14 %' 15-17='15-17 %' 18-20='18-20 %' 

21-9998='21+ %' -9999='All';

run;

proc summary data=deal_review;

class fico pti;

var ever30;

output out=_stats_ mean= n=n_loans;

run;

data _stats_;

set _stats_;

if missing(fico) then fico=-9999;

if missing(pti) then pti=-9999;

run;

PROC SUMMARY Overview

▪ Hashtag

Feel free to use the following hashtag on social media to indicate to all your 

friends how cool you are now

#ProcSummaryRulez

Contact: PaigeMiller at 

communities.sas.com


